Solve Constained minimization using Lagrange Multiplier

Minimize

```
> f := x*y ;
```

$$f := x\,y \tag{1}$$

Constraints

```
> g1 := x^2+y^2 <= 1 ;
```

$$g1 := x^2 + y^2 \le 1 \tag{2}$$

```
> g2 := x^2 <= y ;
```

$$g2 := x^2 \le y \tag{3}$$

Trick: use slack variables to transform inequality into equality constraints.
Drawback: Problem increase in dimension!

Add variable `a` for constraint g1

```
> h1 := rhs(g1) - lhs(g1) - a^2 ;
```

$$h1 := -a^2 - x^2 - y^2 + 1 \tag{4}$$

Add variable `b` for constraint g2

```
> h2 := rhs(g2) - lhs(g2) - b^2 ;
```

$$h2 := -b^2 - x^2 + y \tag{5}$$

Now the problem is 4 dimensional, the original one was 2 dimensional

Step 1, build Lagrangian and non linear system

```
> L := f-lambda1*h1-lambda2*h2 ;
```

$$L := -\left(-a^2 - x^2 - y^2 + 1\right)\lambda 1 - \left(-b^2 - x^2 + y\right)\lambda 2 + x\,y \tag{1.1}$$

Nonlinear system is the gradient of the Lagrangian

```
> EQ1 := diff(L,x) ;
  EQ2 := diff(L,y) ;
  EQ3 := diff(L,a) ;
  EQ4 := diff(L,b) ;
  EQ5 := diff(L,lambda1) ;
  EQ6 := diff(L,lambda2) ;
```

$$EQ1 := 2\,\lambda 1\,x + 2\,\lambda 2\,x + y$$

$$EQ2 := 2\,\lambda 1\,y - \lambda 2 + x$$

$$EQ3 := 2\,a\,\lambda 1$$

$$EQ4 := 2\,b\,\lambda 2$$

$$EQ5 := a^2 + x^2 + y^2 - 1$$

$$EQ6 := b^2 + x^2 - y \tag{1.2}$$

Step 2: solve the nonlinear system

Case a=b=0, i.e. we are on the border of the contraints

```
> EQRED1 := subs( a=0, b=0, [EQ||(1..6)]): <%> ;
```

$$\begin{bmatrix} 2\,\lambda 1\,x + 2\,\lambda 2\,x + y \\ 2\,\lambda 1\,y - \lambda 2 + x \\ 0 \\ 0 \\ x^2 + y^2 - 1 \\ x^2 - y \end{bmatrix} \tag{2.1}$$

```
> SOL112 := solve( EQRED1[1..2], {lambda1,lambda2} ) ;
```

$$SOL112 := \left\{\lambda 1 = -\frac{1}{2}\,\frac{2\,x^2 + y}{x\,(2\,y+1)},\ \lambda 2 = \frac{x^2 - y^2}{x\,(2\,y+1)}\right\} \tag{2.2}$$

Non linear system for x,y

```
> subs( SOL112, EQRED1[5..6] ) ;
```

$$[x^2 + y^2 - 1,\ x^2 - y] \tag{2.3}$$

```
> EQONLYY := EQRED1[5]-EQRED1[6];
```

$$EQONLYY := y^2 + y - 1 \tag{2.4}$$

```
> SOLY := solve( EQONLYY, [y] ) ; evalf(%) ;
```

$$SOLY := \left[\left[y = \frac{1}{2}\sqrt{5} - \frac{1}{2}\right], \left[y = -\frac{1}{2} - \frac{1}{2}\sqrt{5}\right]\right]$$

$$[[y = 0.6180339880],\ [y = -1.618033988]] \tag{2.5}$$

Discard second solution...

Now build alla the solutions

Solution 1:

```
> a_sol1, b_sol1 := 0, 0 ;
```

$$a\_sol1, b\_sol1 := 0, 0 \tag{2.6}$$

```
> y_sol1 := subs(SOLY[1],y);
```

$$y\_sol1 := \frac{1}{2}\sqrt{5} - \frac{1}{2} \tag{2.7}$$

```
> x_sol1 := sqrt(y_sol1) ; evalf(%) ;
```

$$x\_sol1 := \frac{1}{2}\sqrt{-2 + 2\sqrt{5}}$$

$$0.7861513775 \tag{2.8}$$

```
> lambda1_sol1 := simplify(subs(x=x_sol1,y=y_sol1,subs(SOL112,
  lambda1))) ;
```

$$lambda1\_sol1 := -\frac{3}{10}\,\frac{\left(\sqrt{5} - 1\right)\sqrt{5}}{\sqrt{-2 + 2\sqrt{5}}} \tag{2.9}$$

```
> lambda2_sol1 := simplify(subs(x=x_sol1,y=y_sol1,subs(SOL112,
  lambda2))) ;
```

$$lambda2\_sol1 := \frac{2}{5}\,\frac{\left(-2 + \sqrt{5}\right)\sqrt{5}}{\sqrt{-2 + 2\sqrt{5}}} \tag{2.10}$$

Solution 2:

```
> a_sol2, b_sol2 := 0, 0 ;
```
$$a\_sol2, b\_sol2 := 0, 0 \tag{2.11}$$

```
> y_sol2 := subs(SOLY[1],y);
```
$$y\_sol2 := \frac{1}{2}\sqrt{5} - \frac{1}{2} \tag{2.12}$$

```
> x_sol2 := -sqrt(y_sol2) ;
```
$$x\_sol2 := -\frac{1}{2}\sqrt{-2 + 2\sqrt{5}} \tag{2.13}$$

```
> lambda1_sol2 := simplify(subs(x=x_sol2,y=y_sol2,subs(SOLl12,
  lambda1))) ;
```
$$lambda1\_sol2 := \frac{3}{10}\frac{\left(\sqrt{5} - 1\right)\sqrt{5}}{\sqrt{-2 + 2\sqrt{5}}} \tag{2.14}$$

```
> lambda2_sol2 := simplify(subs(x=x_sol2,y=y_sol2,subs(SOLl12,
  lambda2))) ;
```
$$lambda2\_sol2 := -\frac{2}{5}\frac{\left(-2 + \sqrt{5}\right)\sqrt{5}}{\sqrt{-2 + 2\sqrt{5}}} \tag{2.15}$$

Case a<>0 and b<>0 --> lambda1=lambda2=0, i.e. we are INSIDE of BOTH the contraints

```
> EQRED2 := subs( lambda1=0, lambda2=0, [EQ||(1..6)]): <%> ;
```
$$\begin{bmatrix} y \\ x \\ 0 \\ 0 \\ a^2 + x^2 + y^2 - 1 \\ b^2 + x^2 - y \end{bmatrix} \tag{2.16}$$

The solution is easy... (consider only non negative solution for slack variables)

```
> lambda1_sol3, lambda2_sol3 := 0, 0 ;
```
$$lambda1\_sol3, lambda2\_sol3 := 0, 0 \tag{2.17}$$

```
> x_sol3 := 0 ;
  y_sol3 := 0 ;
```
$$x\_sol3 := 0$$
$$y\_sol3 := 0 \tag{2.18}$$

```
> subs( x=0,y=0,EQRED2) ;
```
$$\left[0, 0, 0, 0, a^2 - 1, b^2\right] \tag{2.19}$$

```
> a_sol3, b_sol3 := 1, 0 ;
```
$$a\_sol3, b\_sol3 := 1, 0 \tag{2.20}$$

Case a<>0 and b=0 --> lambda1=0, i.e. we are INSIDE only of the first contraint

```
> EQRED3 := subs( lambda1=0, b=0, [EQ||(1..6)]): <%> ;
```

$$\begin{bmatrix} 2\,\lambda 2\,x + y \\ -\lambda 2 + x \\ 0 \\ 0 \\ a^2 + x^2 + y^2 - 1 \\ x^2 - y \end{bmatrix} \tag{2.21}$$

```
> solve( EQRED3[2], {lambda2} ) ;
  EQRED3bis := subs( %, EQRED3 ) : <%> ;
```

$$\{\lambda 2 = x\}$$

$$\begin{bmatrix} 2\,x^2 + y \\ 0 \\ 0 \\ 0 \\ a^2 + x^2 + y^2 - 1 \\ x^2 - y \end{bmatrix} \tag{2.22}$$

```
> solve( { EQRED3bis[1],EQRED3bis[6]}, {x,y} ) ;
```

$$\{x = 0, y = 0\} \tag{2.23}$$

Solution already considered...

Case a=0 and b<>0 --> lambda2=0, i.e. we are INSIDE only of the second contraint

```
> EQRED4 := subs( lambda2=0, a=0, [EQ||(1..6)]): <%> ;
```

$$\begin{bmatrix} 2\,\lambda 1\,x + y \\ 2\,\lambda 1\,y + x \\ 0 \\ 0 \\ x^2 + y^2 - 1 \\ b^2 + x^2 - y \end{bmatrix} \tag{2.24}$$

```
> SOLX := solve( EQRED4[2], {x} ) ;
```

$$SOLX := \{x = -2\,\lambda 1\,y\} \tag{2.25}$$

```
> EQRED4bis := subs( SOLX, EQRED4 ) : <%> ;
```

(2.26)

$$\begin{bmatrix} -4\,\lambda 1^2\,y + y \\ 0 \\ 0 \\ 0 \\ 4\,\lambda 1^2\,y^2 + y^2 - 1 \\ 4\,\lambda 1^2\,y^2 + b^2 - y \end{bmatrix} \qquad (2.26)$$

```
> subs(y=1, EQRED4bis[1]) ; LAMBDA1 := solve( %, {lambda1} ) ;
```

$$-4\,\lambda 1^2 + 1$$

$$LAMBDA1 := \left\{\lambda 1 = -\frac{1}{2}\right\}, \left\{\lambda 1 = \frac{1}{2}\right\} \qquad (2.27)$$

```
> subs( LAMBDA1[1], EQRED4bis[5] ) ; YSOL := solve(%,{y} ) ;
```

$$2\,y^2 - 1$$

$$YSOL := \left\{y = \frac{1}{2}\sqrt{2}\right\}, \left\{y = -\frac{1}{2}\sqrt{2}\right\} \qquad (2.28)$$

```
> subs(LAMBDA1[1],YSOL[1],EQRED4bis[6]) ;
  BSOL := solve( %, {b}) ;
```

$$b^2 + \frac{1}{2} - \frac{1}{2}\sqrt{2}$$

$$BSOL := \left\{b = \frac{1}{2}\sqrt{-2 + 2\sqrt{2}}\right\}, \left\{b = -\frac{1}{2}\sqrt{-2 + 2\sqrt{2}}\right\} \qquad (2.29)$$

Build the solution

```
> b_sol4       := subs(BSOL[1],b) ;
  lambda1_sol4 := subs(LAMBDA1[1],lambda1) ;
  y_sol4       := subs(YSOL[1],y) ;
```

$$b\_sol4 := \frac{1}{2}\sqrt{-2 + 2\sqrt{2}}$$

$$lambda1\_sol4 := -\frac{1}{2}$$

$$y\_sol4 := \frac{1}{2}\sqrt{2} \qquad (2.30)$$

to be or not to be

Check first solution for second order condition

Compute the gradient of the constraints

```
> gradH1 := <diff(h1,x)|diff(h1,y)|diff(h1,a)|diff(h1,b)>;
  gradH2 := <diff(h2,x)|diff(h2,y)|diff(h2,a)|diff(h2,b)>;
```

$$gradH1 := \begin{bmatrix} -2\,x & -2\,y & -2\,a & 0 \end{bmatrix}$$

$$gradH2 := \begin{bmatrix} -2x & 1 & 0 & -2b \end{bmatrix} \quad (3.1)$$

```
> H := < gradH1, gradH2 > ;
  H1 := subs( x=x_sol1, y=y_sol1, a=a_sol1, b=b_sol1, H ) ;
```

$$H := \begin{bmatrix} -2x & -2y & -2a & 0 \\ -2x & 1 & 0 & -2b \end{bmatrix}$$

$$H1 := \begin{bmatrix} -\sqrt{-2+2\sqrt{5}} & -\sqrt{5}+1 & 0 & 0 \\ -\sqrt{-2+2\sqrt{5}} & 1 & 0 & 0 \end{bmatrix} \quad (3.2)$$

```
> NS := op(LinearAlgebra[NullSpace](H1)) ;
```

$$NS := \begin{bmatrix} 0 \\ 0 \\ 0 \\ 1 \end{bmatrix}, \begin{bmatrix} 0 \\ 0 \\ 1 \\ 0 \end{bmatrix} \quad (3.3)$$

```
> Z := alpha*NS[1]+beta*NS[2] ;
```

$$Z := \begin{bmatrix} 0 \\ 0 \\ \beta \\ \alpha \end{bmatrix} \quad (3.4)$$

```
> H1.Z ;
```

$$\begin{bmatrix} 0 \\ 0 \end{bmatrix} \quad (3.5)$$

Compute the Hessian of the Lagrangian

```
> HessL := <<diff(L,x,x),
               diff(L,x,y),
                 diff(L,x,a),
                   diff(L,x,b)>|
            <diff(L,y,x),
               diff(L,y,y),
                 diff(L,y,a),
                   diff(L,y,b)>|
            <diff(L,a,x),
               diff(L,a,y),
                 diff(L,a,a),
                   diff(L,a,b)>|
            <diff(L,b,x),
               diff(L,b,y),
                 diff(L,b,a),
                   diff(L,b,b)>>;
```

(3.6)

$$HessL := \begin{bmatrix} 2\lambda 1 + 2\lambda 2 & 1 & 0 & 0 \\ 1 & 2\lambda 1 & 0 & 0 \\ 0 & 0 & 2\lambda 1 & 0 \\ 0 & 0 & 0 & 2\lambda 2 \end{bmatrix} \quad (3.6)$$

```
> HessL0 := subs( x=x_sol1, y=y_sol1,
                  a=a_sol1, b=b_sol1,
                  lambda1=0,lambda2=0,HessL ) ;
```

$$HessL0 := \begin{bmatrix} 0 & 1 & 0 & 0 \\ 1 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{bmatrix} \quad (3.7)$$

```
> LinearAlgebra[Transpose](Z).HessL0.Z ;
```

$$0 \quad (3.8)$$

Check second solution for second order condition

Compute the gradient of the constraints

```
> H2 := subs( x=x_sol2, y=y_sol2, a=a_sol2, b=b_sol2, H ) ;
```

$$H2 := \begin{bmatrix} \sqrt{-2 + 2\sqrt{5}} & -\sqrt{5} + 1 & 0 & 0 \\ \sqrt{-2 + 2\sqrt{5}} & 1 & 0 & 0 \end{bmatrix} \quad (4.1)$$

```
> NS := op(LinearAlgebra[NullSpace](H2)) ;
```

$$NS := \begin{bmatrix} 0 \\ 0 \\ 0 \\ 1 \end{bmatrix}, \begin{bmatrix} 0 \\ 0 \\ 1 \\ 0 \end{bmatrix} \quad (4.2)$$

```
> Z := alpha*NS[1]+beta*NS[2] ;
```

$$Z := \begin{bmatrix} 0 \\ 0 \\ \beta \\ \alpha \end{bmatrix} \quad (4.3)$$

```
> H2.Z ;
```

$$\begin{bmatrix} 0 \\ 0 \end{bmatrix} \quad (4.4)$$

Compute the Hessian of the Lagrangian

```
> HessL2 := subs( x=x_sol2, y=y_sol2,
                  a=a_sol2, b=b_sol2,
```

```
                lambda1=lambda1_sol2,
                lambda2=lambda2_sol2,HessL ) ;
```

$$HessL2 := \left[\left[\frac{3}{5}\,\frac{\left(\sqrt{5}-1\right)\sqrt{5}}{\sqrt{-2+2\sqrt{5}}}-\frac{4}{5}\,\frac{\left(-2+\sqrt{5}\right)\sqrt{5}}{\sqrt{-2+2\sqrt{5}}}, 1, 0, 0\right],\right. \tag{4.5}$$

$$\left[1, \frac{3}{5}\,\frac{\left(\sqrt{5}-1\right)\sqrt{5}}{\sqrt{-2+2\sqrt{5}}}, 0, 0\right],$$

$$\left[0, 0, \frac{3}{5}\,\frac{\left(\sqrt{5}-1\right)\sqrt{5}}{\sqrt{-2+2\sqrt{5}}}, 0\right],$$

$$\left.\left[0, 0, 0, -\frac{4}{5}\,\frac{\left(-2+\sqrt{5}\right)\sqrt{5}}{\sqrt{-2+2\sqrt{5}}}\right]\right]$$

```
> LinearAlgebra[Transpose](Z).HessL2.Z ; evalf(%) ;
```

$$\frac{3}{5}\,\frac{\beta^2\left(\sqrt{5}-1\right)\sqrt{5}}{\sqrt{-2+2\sqrt{5}}}-\frac{4}{5}\,\frac{\alpha^2\left(-2+\sqrt{5}\right)\sqrt{5}}{\sqrt{-2+2\sqrt{5}}}$$

$$1.054732753\,\beta^2-0.2685813745\,\alpha^2 \tag{4.6}$$